Equações Diferenciais Ordinárias de Primeira Ordem
Problemas Resolvidos

Fernando Deeke Sasse
Departmento de Matemática
CCT - UDESC

1 Determinar a curva que passa pelo ponto (2,0) e possui em cada um de seus pontos o coeficiente angular $\frac{x}{4\,y}$.

Solução:

```
> restart:
> eq:=diff(y(x),x)=x/(4*y(x));
```

$$eq := \frac{d}{dx}\, y(x) = \frac{1}{4}\, \frac{x}{y(x)} \tag{1.1}$$

```
> dsolve({eq,y(0)=2},y(x));
```

$$y(x) = \frac{\sqrt{x^2+16}}{2} \tag{1.2}$$

```
> ;
```

2 Resolver o problema de valor inicial $2y' = \frac{y}{x} - \frac{x}{y^2}$ $\quad y(1) = 1$ Em que região do plano xy a solução garantidamente existe e é única ?

Solução:

```
> restart:
> eq1:=diff(y(x),x)*2=y(x)/x-x/(y(x)^2);
```

$$eq1 := 2\left(\frac{d}{dx}\, y(x)\right) = \frac{y(x)}{x} - \frac{x}{y(x)^2} \tag{2.1}$$

Esta é uma equação de Bernoulli. Sua forma padrão é

```
> eq2:= diff(y(x),x)-y(x)/(2*x)=x/(2*y(x)^2);
```

$$eq2 := \frac{d}{dx}\, y(x) - \frac{1}{2}\, \frac{y(x)}{x} = \frac{1}{2}\, \frac{x}{y(x)^2} \tag{2.2}$$

Multiplicando toda a equação por y^2 temos

```
> eq3:=expand(eq2*y(x)^2);
```

$$eq3 := y(x)^2 \left(\frac{d}{dx} y(x)\right) - \frac{1}{2} \frac{y(x)^3}{x} = \frac{1}{2} x \qquad (2.3)$$

```
> eq4:= y(x)^3=u(x);
```

$$eq4 := y(x)^3 = u(x) \qquad (2.4)$$

```
> diff(eq4,x);
```

$$3\, y(x)^2 \left(\frac{d}{dx} y(x)\right) = \frac{d}{dx} u(x) \qquad (2.5)$$

```
> eq5:=diff(u(x),x)/3-u(x)/(2*x)=x/2;
```

$$eq5 := \frac{1}{3} \frac{d}{dx} u(x) - \frac{1}{2} \frac{u(x)}{x} = \frac{1}{2} x \qquad (2.6)$$

Esta é uma equação linear que tem como solução:

```
> sol1:=dsolve(eq5,u(x));
```

$$sol1 := u(x) = 3\, x^2 + x^{3/2}\, \_C1 \qquad (2.7)$$

```
> sol2:=subs(u(x)=`y`^3,sol1);
```

$$sol2 := y^3 = 3\, x^2 + x^{3/2}\, \_C1 \qquad (2.8)$$

```
> subs({x = 1, y = 1}, sol2);
```

$$1 = 3 + \_C1 \qquad (2.9)$$

```
> _C1 := solve(%,_C1);
```

$$\_C1 := -2 \qquad (2.10)$$

> *sol2*

$$y^3 = 3\, x^2 - 2\, x^{3/2} \qquad (2.11)$$

Notemos que quando $x = 0$ esta solução implica $y = 0$ para qualquer constante C1. Ou seja, o problema de valor inicial
não pode ser satisfeito. De fato, como $\frac{y(x)}{x} - \frac{x}{y(x)^2}$ possui singularidades em $x = 0$ e $y = 0$, o teorema de existência e unicidade não garante a unicidade de soluções em pontos correspondentes a $x = 0$ ou $y = 0$.

```
> ;
```

3 A massa de um determinado material radioativo decresce a uma taxa proporcional à quantidade presente em qualquer tempo. Sua meia-vida é de 4,5 horas. Se uma massa de 3g deste material está presente inicialmente, quanto tempo levará para 99% do material desaparecer ?

Solução:

```
> restart:
> eq1:=diff(M(t),t)=k*M(t);
```

$$eq1 := \frac{d}{dt} M(t) = k\, M(t) \qquad (3.1)$$

```
> sol:=dsolve({eq1,M(0)=3},M(t));
```

$$sol := M(t) = 3\,e^{k\,t} \quad (3.2)$$

```
> m:=solve(sol,M(t));
```

$$m := 3\,e^{k\,t} \quad (3.3)$$

```
> m:=unapply(m,t);
```

$$m := t \rightarrow 3\,e^{k\,t} \quad (3.4)$$

```
> eq2:=m(0)/2=m(4.5);
```

$$eq2 := \frac{3}{2} = 3\,e^{4.5\,k} \quad (3.5)$$

```
> k:=solve(eq2,k);
```

$$k := -0.1540327068 \quad (3.6)$$

```
> eq3:=0.01*m(0)=m(T);
```

$$eq3 := 0.03 = 3\,e^{-0.1540327068\,T} \quad (3.7)$$

```
> T:=solve(eq3,T);
```

$$T := 29.89735285 \quad (3.8)$$

```
> ;
```

4 Um termômetro é retirado de dentro de uma sala e colocado do lado de fora, onde a temperatura é de 10 C. Após 1 min. o termômetro marcava 20 C. Após 5 minutos, 16 C. Qual a temperatura da sala ?.

Solução:

```
> restart:
> eq1:=diff(T(t),t)=-k*(T(t)-10);
```

$$eq1 := \frac{d}{dt}\,T(t) = -k\,(T(t) - 10) \quad (4.1)$$

```
> s1:=dsolve(eq1,T(t));
```

$$s1 := T(t) = 10 + e^{-k\,t}\,\_C1 \quad (4.2)$$

```
> TT:=solve(s1,T(t));
```

$$TT := 10 + e^{-k\,t}\,\_C1 \quad (4.3)$$

```
> TT:=unapply(%,t);
```

$$TT := t \rightarrow 10 + e^{-k\,t}\,\_C1 \quad (4.4)$$

```
> e1:=T0=TT(0);
```

$$e1 := T0 = 10 + \_C1 \quad (4.5)$$

```
> _C1:=solve(e1,_C1);
```

$$\_C1 := T0 - 10 \quad (4.6)$$

```
> s1;
```

$$T(t) = 10 + e^{-k\,t}\,(T0 - 10) \quad (4.7)$$

```
> e2:=20=TT(1);
```
$$e2 := 20 = 10 + e^{-k} (T0 - 10) \quad \textbf{(4.8)}$$

```
> e3:=16=TT(5);
```
$$e3 := 16 = 10 + e^{-5k} (T0 - 10) \quad \textbf{(4.9)}$$

```
> fsolve({e2,e3},{k,T0});
```
$$\{T0 = 21.36219366, k = 0.1277064059\} \quad \textbf{(4.10)}$$

```
> ;
```

5 (3 pontos) Determinar a solução das equações

(a) $y'(3x^2 - y^2) - 2xy = 0$ (b) $y'' + y' \tan(x) = \cos(2x)$

Solução:

(a)

```
> restart:
> M:=-2*x*y;
```
$$M := -2xy \quad \textbf{(5.1)}$$

```
> N:=3*x^2-y^2;
```
$$N := 3x^2 - y^2 \quad \textbf{(5.2)}$$

Como

```
> diff(M,y);diff(N,x);
```
$$-2x \quad \textbf{(5.3)}$$
$$6x$$

vemos que a equação não é exata. Busquemos um fator integrante mu. Tal função deve ser tal que

```
> eq1:=diff(mu(x,y)*M,y)=diff(mu(x,y)*N,x);
```
$$eq1 := -2\left(\frac{\partial}{\partial y}\mu(x,y)\right)xy - 2\mu(x,y)x = \left(\frac{\partial}{\partial x}\mu(x,y)\right)(3x^2 - y^2) + 6\mu(x,y)x \quad \textbf{(5.4)}$$

Supondo $\mathrm{mu} = \mathrm{mu}(x)$ temos

```
> mu(x,y):=mu(x);
```
$$\mu(x,y) := \mu(x) \quad \textbf{(5.5)}$$

```
> eq1;
```
$$-2\mu(x)x = \left(\frac{d}{dx}\mu(x)\right)(3x^2 - y^2) + 6\mu(x)x \quad \textbf{(5.6)}$$

Como a equação depende de y, tal escolha é inconsistente. Tentamos então $\mathrm{mu} = \mathrm{mu}(y)$. Temos então

```
> mu(x,y):=mu(y);
```
$$\mu(x,y) := \mu(y) \quad \textbf{(5.7)}$$

```
> eq1;
```
$$-2\left(\frac{d}{dy}\mu(y)\right)xy - 2\mu(y)x = 6\mu(y)x \quad \textbf{(5.8)}$$

ou

```
> eq2:=simplify((rhs(eq1)-lhs(eq1))/x)=0;
```

$$eq2 := 8\,\mu(y) + 2\left(\frac{d}{dy}\,\mu(y)\right)y = 0 \tag{5.9}$$

```
> eq3:=subs(mu(y)=MM(y),eq2);
```

$$eq3 := 8\,MM(y) + 2\left(\frac{d}{dy}\,MM(y)\right)y = 0 \tag{5.10}$$

```
> dsolve(eq3,MM(y));
```

$$MM(y) = \frac{\_C1}{y^4} \tag{5.11}$$

```
> mu:=op(2,%);
```

$$\mu := \frac{\_C1}{y^4} \tag{5.12}$$

Basta escolher

```
> _C1:=1;
```

$$\_C1 := 1 \tag{5.13}$$

A EDO

```
> eq4:=Nt*diff(y(x),x)+Mt=0;
```

$$eq4 := Nt\left(\frac{d}{dx}\,y(x)\right) + Mt = 0 \tag{5.14}$$

onde

```
> Nt:=mu*N;Mt:=mu*M;
```

$$Nt := \frac{3\,x^2 - y^2}{y^4} \tag{5.15}$$

$$Mt := -\frac{2\,x}{y^3}$$

agora é exata, pois

```
> diff(Nt,x);diff(Mt,y);
```

$$\frac{6\,x}{y^4} \tag{5.16}$$

$$\frac{6\,x}{y^4}$$

A EDO exata deve ser da forma $dF = \left(\frac{\partial}{\partial x}\,F(x,y)\right)dx + \left(\frac{\partial}{\partial y}\,F(x,y)\right)dy = 0$, com solução $F = const$. Temos então

```
> ee1:=diff(F(x,y),x)=Mt;
```

$$ee1 := \frac{\partial}{\partial x}\,F(x,y) = -\frac{2\,x}{y^3} \tag{5.17}$$

```
> ee2:=diff(F(x,y),y)=Nt;
```

$$ee2 := \frac{\partial}{\partial y} F(x,y) = \frac{3x^2 - y^2}{y^4} \quad \textbf{(5.18)}$$

Integrando a primeira equação obtemos

```
> EE1:=subs(F(x,y)=f(x),ee1);
```

$$EE1 := \frac{d}{dx} f(x) = -\frac{2x}{y^3} \quad \textbf{(5.19)}$$

```
> dsolve(EE1,f(x));
```

$$f(x) = -\frac{x^2}{y^3} + \_C2 \quad \textbf{(5.20)}$$

```
> ff:=op(2,%);
```

$$ff := -\frac{x^2}{y^3} + \_C2 \quad \textbf{(5.21)}$$

```
> _C2:=c2(y);
```

$$\_C2 := c2(y) \quad \textbf{(5.22)}$$

```
> EE2:=subs(F(x,y)=ff,ee2);
```

$$EE2 := \frac{\partial}{\partial y}\left(-\frac{x^2}{y^3} + c2(y)\right) = \frac{3x^2 - y^2}{y^4} \quad \textbf{(5.23)}$$

```
> eq5:=simplify(lhs(EE2)-rhs(EE2))=0;
```

$$eq5 := \frac{\left(\frac{d}{dy} c2(y)\right) y^2 + 1}{y^2} = 0 \quad \textbf{(5.24)}$$

```
> eq6:=dsolve(eq5,c2(y));
```

$$eq6 := c2(y) = \frac{1}{y} + \_C3 \quad \textbf{(5.25)}$$

```
> _C3:=0:
> c2(y):=rhs(eq6);
```

$$c2(y) := \frac{1}{y} \quad \textbf{(5.26)}$$

Portanto, a solução é

```
> ff = const;
```

$$-\frac{x^2}{y^3} + \frac{1}{y} = const \quad \textbf{(5.27)}$$

Podemos verificar que a derivada total de ff nos dá a EDO original:

```
> simplify(y^4*(diff(ff,y)*diff(y(x),x)+diff(ff,x)))=0;
```

$$3\left(\frac{d}{dx} y(x)\right) x^2 - \left(\frac{d}{dx} y(x)\right) y^2 - 2xy = 0 \quad \textbf{(5.28)}$$

(b)

```
> restart:
```

```
> eq1:=diff(y(x),x$2)+diff(y(x),x)*tan(x)=cos(2*x);
```

$$eq1 := \frac{d^2}{dx^2} y(x) + \left(\frac{d}{dx} y(x)\right) \tan(x) = \cos(2x) \qquad (5.29)$$

```
> eq2:=subs({diff(y(x),x$2)=diff(p(x),x),diff(y(x),x)=p(x) },eq1)
;
```

$$eq2 := \frac{d}{dx} p(x) + p(x) \tan(x) = \cos(2x) \qquad (5.30)$$

```
> Y:=u*v;
```

$$Y := u\, v \qquad (5.31)$$

```
> v:=exp(-Int(tan(x),x));
```

$$v := e^{-\left(\int \tan(x)\, dx\right)} \qquad (5.32)$$

```
> v:=value(%);
```

$$v := \cos(x) \qquad (5.33)$$

```
> u:=Int(cos(2*x)/v,x)+C1;
```

$$u := \int \frac{\cos(2x)}{\cos(x)}\, dx + C1 \qquad (5.34)$$

```
> u:=value(%);
```

$$u := 2\sin(x) - \ln(\sec(x) + \tan(x)) + C1 \qquad (5.35)$$

```
> p:=u*v;
```

$$p := (2\sin(x) - \ln(\sec(x) + \tan(x)) + C1) \cos(x) \qquad (5.36)$$

```
> Y:=simplify(Int(p,x))+C2;
```

$$Y := \int \left(2\sin(x) - \ln\left(\frac{1+\sin(x)}{\cos(x)}\right) + C1\right) \cos(x)\, dx + C2 \qquad (5.37)$$

Tal integral só tem soluções complexas:

```
> simplify(value(Y),exp);
```

$$-\frac{1}{4} e^{2Ix} - \frac{1}{4} e^{-2Ix} + \frac{1}{2} I \ln\left(\frac{2e^{xI} - I e^{2Ix} + I}{e^{2Ix}+1}\right) e^{xI} - \ln(e^{2Ix}+1) - \frac{1}{2} I \ln\left(\frac{2e^{xI} - I e^{2Ix} + I}{e^{2Ix}+1}\right) e^{-Ix} + \ln(e^{xI}) - \frac{1}{2} I\, C1\, e^{xI} + \frac{1}{2} I\, C1\, e^{-Ix} + C2 \qquad (5.38)$$

De forma direta:

```
> dsolve(eq1,y(x));
```

$$y(x) = -\frac{1}{2}\cos(2x) + \frac{1}{2} I \ln\left(-\frac{-2e^{xI} + e^{2Ix} I - I}{e^{2Ix}+1}\right) e^{xI} - \ln(e^{2Ix}+1) - \frac{1}{2} I \ln\left(-\frac{-2e^{xI} + e^{2Ix} I - I}{e^{2Ix}+1}\right) e^{-Ix} + \ln(e^{xI}) + \sin(x)\, \_C1 + \_C2 \qquad (5.39)$$

6 Determinar a equação diferencial que determina o movimento vertical de um corpo sob a ação do campo gravitacional da Terra. A equação diferencial deve ser dada em termos da velocidade, do raio da Terra R, da altura do corpo em relação à superfície x, da aceleração da gravidade na superfície da Terra g e da massa m do corpo. A força sobre o corpo é dada pela lei de Newton da gravitação universal, $F = -\frac{G M m}{r^2}$ onde G é a constante universal da gravitação, M é a massa da terra e r a distância do corpo ao centro da Terra (supomos positiva a direção associada a r crescente). Suponha que a atmosfera exerce uma força dissipativa sobre o corpo que é proporcional à sua velocidade. Não é necessário resolver a equação diferencial.

```
> restart:
```

Se gamma é a constante dissipativa da atmosfera, temos

```
> eq1:=m*diff(v(x),x)*v(x)=-G*M*m/(x+R)^2-gamma*v(x);
```

$$eq1 := m \left(\frac{d}{dx} v(x)\right) v(x) = -\frac{G M m}{(x+R)^2} - \gamma v(x) \tag{6.1}$$

Na superfície,:

```
> ee1:=m*g=M*m*G/R^2;
```

$$ee1 := m g = \frac{M m G}{R^2} \tag{6.2}$$

```
> ee1/m*R^2;
```

$$R^2 g = M G \tag{6.3}$$

Com isso eliminamos a massa da Terra em termos do raio e da aceleração local da gravidade (calculáveis desde a antiguidade). Nossa EDO torna-se, então.

```
> eq2:=diff(v(x),x)*v(x)=-R^2*g/(x+R)^2-gamma*v(x)/m;
```

$$eq2 := \left(\frac{d}{dx} v(x)\right) v(x) = -\frac{R^2 g}{(x+R)^2} - \frac{\gamma v(x)}{m} \tag{6.4}$$

```
> ;
```